MATRIX REPRESENTATIONS OF SOME FINITE GROUPS

LOUIS ESSER AND JENNIFER LI

1. Introduction

This note contains some supporting code for the paper "Hypersurfaces with large automorphism groups." We used both MAGMA and GAP to calculate invariant polynomials or their degrees, though many of the results of these calculations were to confirm existing results in the literature.

The main tools are as follows: in MAGMA, one can create a matrix group object and then use the function InvariantsOfDegree to find invariant polynomials. We can do similarly for semi-invariant polynomials because semi-invariants are invariants of normal subgroups with abelian quotient (see Lemma 1.3 of the main paper).

Alternatively, GAP has databases of representations of finite groups. The function MolienSeries computes the power series whose coefficients are the dimensions of each graded piece of the invariant ring of your given representation.

The sections below provide the implementations of these functions for various computations performed in the paper. The references for the matrix generators are given in the main paper.

2. $N = 2$

2.1. **Binary Icosahedral Group $2.A_5$.**

In appropriate coordinates, the generators for the binary icosahedral group $2.A_5$ of order 120 are

$$M_1 = \begin{pmatrix} -\epsilon^3 & 0 \\ 0 & -\epsilon^2 \end{pmatrix}, M_2 = \frac{1}{\sqrt{5}} \begin{pmatrix} -\epsilon + \epsilon^4 & \epsilon^2 - \epsilon^3 \\ \epsilon^2 - \epsilon^3 & \epsilon - \epsilon^4, \end{pmatrix},$$

where ϵ is a primitive 5th root of unity.

MAGMA implementation:

```
K<z> := CyclotomicField(5);
M1 := Matrix(K,2,2,[-z^3,0, 0,-z^2]);
M2 := 1/(2*z + 2*z^4 + 1)*Matrix(K,2,2,[-z+z^4, z^2-z^3,  z^2-z^3, z-z^4]);
G := MatrixGroup<2,K | M1,M2>;
InvariantsOfDegree(G,12);
```

The code above returns the smallest degree invariant polynomial of this group: $xy(x^{10} + 11x^5y^5 + y^{10})$.

Alternatively, we could use GAP:

```
gap> G := AtlasGroup("2.A5");
Group([ (1,2,5,4)(3,6,8,7)(9,13,11,14)(10,15,12,16)(17,19,18,20)(21,24,23,22),
(1,3,2)(4,5,8)(6,9,10)(7,11,12)(13,16,17)(14,15,18)(19,21,22)(20,23,24) ])
```

The `AtlasGroup` function contains many common groups, such as the binary icosahedral group.

```
gap> Display(Irr(G));
[1,             1,                1,  1,  1,  1,               1,  1,               1],
[2,    E(5)+E(5)^4,  E(5)^2+E(5)^3, -1,  0,  1,   -E(5)-E(5)^4, -2, -E(5)^2-E(5)^3],
[2,  E(5)^2+E(5)^3,    E(5)+E(5)^4, -1,  0,  1, -E(5)^2-E(5)^3, -2,   -E(5)-E(5)^4],
[3, -E(5)^2-E(5)^3,   -E(5)-E(5)^4,  0, -1,  0, -E(5)^2-E(5)^3,  3,   -E(5)-E(5)^4],
[3,   -E(5)-E(5)^4, -E(5)^2-E(5)^3,  0, -1,  0,   -E(5)-E(5)^4,  3, -E(5)^2-E(5)^3],
[4,            -1,               -1,  1,  0,  1,              -1,  4,              -1],
[4,            -1,               -1,  1,  0, -1,               1, -4,               1],
[5,             0,                0, -1,  1, -1,               0,  5,               0],
[6,             1,                1,  0,  0,  0,              -1, -6,              -1]
```

This command gives you the character table of G, where each row corresponds to an irreducible representation of G (over $\mathbb{C}$) and each column the trace of the matrix for one conjugacy class of elements of G. The first column is the trace of the identity matrix, so it tells you the dimension of the representation. So G has 9 irreducible representations of dimensions $1, 2, 2, 3, 3, 4, 4, 5$, and 6. In this case, we're interested in the embeddings $G \subset \mathrm{GL}_2(\mathbb{C})$, so we'll look at the irreducible representations of dimension 2 (there are two of them, but they have the same degrees of invariants).

```
gap> List([ 0 .. 30 ], i -> ValueMolienSeries( MolienSeries(Irr(G)[2]),i));
[1,0,0,0,0,0,0,0,0,0,0,0,1,0,0,0,0,0,0,0,1,0,0,0,1,0,0,0,0,0,1]
```

This returns the first 30 coefficients of the Molien series of this representation. It confirms that the first (non-constant) invariant polynomial occurs in degree 12.

2.2. **Binary Octahedral and Tetrahedral Groups.** In appropriate coordinates, the generators for the binary octahedral group $2.S_4$ of order 48 are

$$M_1 = \frac{1}{2}\begin{pmatrix} 1+i & 1+i \\ -1+i & 1-i \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix}, M_3 = \begin{pmatrix} \epsilon & 0 \\ 0 & \epsilon^{-1} \end{pmatrix},$$

where ϵ is a primitive 8th root of unity.

The binary tetrahedral group $2.A_4$ is a subgroup of the above group of index 2, which is generated by the first two matrices M_1, M_2.

`MAGMA` implementation:

```
K<z> := CyclotomicField(8);
M1 := 1/2*Matrix(K,2,2,[1+z^2,1+z^2,  -1+z^2,1-z^2]);
M2 := Matrix(K,2,2,[0,1, -1,0]);
M3 := Matrix(K,2,2, [z,0, 0,z^7]);
G := MatrixGroup<2,K | M1,M2,M3>; //binary octahedral group 2.S4
```

```
H := MatrixGroup<2,K | M1,M2>; //binary tetrahedral group 2.A4
```

The only normal subgroup of $2.S_4$ with abelian quotient is $2.A_4 \subset 2.S_4$, so a semi-invariant polynomial of $2.S_4$ must actually be an invariant polynomial of $2.A_4$. MAGMA confirms that the smallest degree invariant polynomial of $2.A_4$ has degree 6, and equals $x_1^5x_2 - x_1x_2^5$.

The only normal subgroup of $2.A_4$ with abelian quotient is the preimage of the normal Klein four-group in $A_4 \subset \mathrm{PGL}_2(\mathbb{C})$, so a semi-invariant polynomial of $2.A_4$ must actually be an invariant polynomial of this subgroup. MAGMA confirms that the smallest degree invariant polynomial for this subgroup has degree 4.

2.3. **Exceptional Hypersurfaces Constructed Using $N = 2$ Groups.** Just to doublecheck that our examples are really correct, let's construct the groups $\mathrm{Lin}(f)$ explicitly in each case.

Example 2.1. Let $X \subset \mathbb{P}^3$ be the degree 6 surface defined by the following equation:

$$x_0^5x_1 - x_0x_1^5 + x_2^5x_3 - x_2x_3^5 = 0.$$

We define the following 2×2 matrices:

$$M_1 = \frac{1}{2}\begin{pmatrix} 1+i & 1+i \\ -1+i & 1-i \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix}, M_3 = \begin{pmatrix} \epsilon & 0 \\ 0 & \epsilon^{19} \end{pmatrix},$$

where ϵ is a primitive 24th root of unity with $\epsilon^6 = i$.

Let the group $G \subset \mathrm{GL}_4(\mathbb{C})$ be generated by block matrices of the form

$$\begin{pmatrix} M_i & 0 \\ 0 & M_j \end{pmatrix},$$

for $i, j \in \{1, 2, 3\}$ and the block matrix

$$\begin{pmatrix} 0 & I_2 \\ I_2 & 0 \end{pmatrix}.$$

MAGMA Implementation:

```
K<y> := CyclotomicField(24);
z := y^3;
M1 := 1/2*Matrix(K,2,2,[1+z^2,1+z^2,  -1+z^2,1-z^2]);
M2 := Matrix(K,2,2,[0,1, -1,0]);
M3 := Matrix(K,2,2, [y,0, 0,y^19]);
B1 := DiagonalJoin(M1,M1);
B2 := DiagonalJoin(M1,M2);
B3 := DiagonalJoin(M1,M3);
B4 := DiagonalJoin(M2,M1);
B5 := DiagonalJoin(M2,M2);
B6 := DiagonalJoin(M2,M3);
B7 := DiagonalJoin(M3,M1);
```

```
B8 := DiagonalJoin(M3,M2);
B9 := DiagonalJoin(M3,M3);
B10 := Matrix(K,4,4,[0,0,1,0, 0,0,0,1, 1,0,0,0, 0,1,0,0]);
G := MatrixGroup<4,K | B1,B2,B3,B4,B5,B6,B7,B8,B9,B10>;
```

This group G has order 41472 and equals $\mathrm{Lin}(f)$, where f is the polynomial above defining the surface.

Example 2.2. Let $X \subset \mathbb{P}^3$ be the degree 12 surface defined by the following equation:

$$x_0x_1(x_0^{10} + 11x_0^5x_1^5 + x_1^{10}) + x_2x_3(x_2^{10} + 11x_2^5x_3^5 + x_3^{10}) = 0.$$

We define the following 2×2 matrices:

$$M_1 = \begin{pmatrix} \zeta & 0 \\ 0 & \zeta^{49} \end{pmatrix}, M_2 = \frac{1}{\sqrt{5}} \begin{pmatrix} -\epsilon + \epsilon^4 & \epsilon^2 - \epsilon^3 \\ \epsilon^2 - \epsilon^3 & \epsilon - \epsilon^4, \end{pmatrix},$$

where ζ is a primitive 60th root of unity and $\epsilon = \zeta^{12}$.

Let the group $G \subset \mathrm{GL}_4(\mathbb{C})$ be generated by block matrices of the form

$$\begin{pmatrix} M_i & 0 \\ 0 & M_j \end{pmatrix},$$

for $i, j \in \{1, 2\}$ and the block matrix

$$\begin{pmatrix} 0 & I_2 \\ I_2 & 0 \end{pmatrix}.$$

MAGMA Implementation:

```
K<y> := CyclotomicField(60);
z := y^12;
M1 := Matrix(K,2,2,[y,0, 0,y^49]);
M2 := 1/(2*z + 2*z^4 + 1)*Matrix(K,2,2,[-z+z^4, z^2-z^3,  z^2-z^3, z-z^4]);
G := MatrixGroup<2,K | M1,M2>;
B1 := DiagonalJoin(M1,M1);
B2 := DiagonalJoin(M1,M2);
B3 := DiagonalJoin(M2,M1);
B4 := DiagonalJoin(M2,M2);
B5 := Matrix(K,4,4,[0,0,1,0, 0,0,0,1, 1,0,0,0, 0,1,0,0]);
G := MatrixGroup<4,K | B1,B2,B3,B4,B5>;
```

This group G has order 1036800 and equals $\mathrm{Lin}(f)$, where f is the polynomial above defining the surface.

Example 2.3. Let $X \subset \mathbb{P}^5$ be the degree 12 hypersurface defined by the equation

$$x_0x_1(x_0^{10} + 11x_0^5x_1^5 + x_1^{10}) + x_2x_3(x_2^{10} + 11x_2^5x_3^5 + x_3^{10}) + x_4x_5(x_4^{10} + 11x_4^5x_5^5 + x_5^{10}) = 0.$$

We define the following 2×2 matrices:

$$M_1 = \begin{pmatrix} \zeta & 0 \\ 0 & \zeta^{49} \end{pmatrix}, M_2 = \frac{1}{\sqrt{5}} \begin{pmatrix} -\epsilon + \epsilon^4 & \epsilon^2 - \epsilon^3 \\ \epsilon^2 - \epsilon^3 & \epsilon - \epsilon^4, \end{pmatrix},$$

where ζ is a primitive 60th root of unity and $\epsilon = \zeta^{12}$.

Let the group $G \subset \mathrm{GL}_6(\mathbb{C})$ be generated by block matrices of the form

$$\begin{pmatrix} M_i & 0 & 0 \\ 0 & M_j & 0 \\ 0 & 0 & M_k \end{pmatrix},$$

for $i, j \in \{1, 2\}$ and the block matrices

$$\begin{pmatrix} 0 & I_2 & 0 \\ I_2 & 0 & 0 \\ 0 & 0 & I_2 \end{pmatrix}, \begin{pmatrix} 0 & I_2 & 0 \\ 0 & 0 & I_2 \\ I_2 & 0 & 0 \end{pmatrix}.$$

`MAGMA` Implementation:

```
K<y> := CyclotomicField(60);
z := y^12;
M1 := Matrix(K,2,2,[y,0, 0,y^49]);
M2 := 1/(2*z + 2*z^4 + 1)*Matrix(K,2,2,[-z+z^4, z^2-z^3,  z^2-z^3, z-z^4]);
G := MatrixGroup<2,K | M1,M2>;
B1 := DiagonalJoin(M1,DiagonalJoin(M1,M1));
B2 := DiagonalJoin(M1,DiagonalJoin(M1,M2));
B3 := DiagonalJoin(M1,DiagonalJoin(M2,M1));
B4 := DiagonalJoin(M1,DiagonalJoin(M2,M2));
B5 := DiagonalJoin(M2,DiagonalJoin(M1,M1));
B6 := DiagonalJoin(M2,DiagonalJoin(M1,M2));
B7 := DiagonalJoin(M2,DiagonalJoin(M2,M1));
B8 := DiagonalJoin(M2,DiagonalJoin(M2,M2));
B9 := Matrix(K,6,6,[0,0,1,0,0,0, 0,0,0,1,0,0, 1,0,0,0,0,0, 0,1,0,0,0,0,
0,0,0,0,1,0, 0,0,0,0,0,1]);
B10 := Matrix(K,6,6,[0,0,1,0,0,0, 0,0,0,1,0,0, 0,0,0,0,1,0, 0,0,0,0,0,1,
1,0,0,0,0,0, 0,1,0,0,0,0]);
G := MatrixGroup<6,K | B1,B2,B3,B4,B5,B6,B7,B8,B9,B10>;
```

This group G has order 2239488000 and equals $\mathrm{Lin}(f)$, where f is the polynomial above defining the fourfold.

3. $N = 3$

For $N = 3$, we are interested in primitive subgroups of $\mathrm{GL}_3(\mathbb{C})$ which have $[G : Z(G)] > 3! \cdot 3^{3-1} = 54$.

3.1. **Three-dimensional Representation of A_5.** In appropriate coordinates, the generators for the subgroup $A_5 \subset \mathrm{GL}_3(\mathbb{C})$ are

$$M_1 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos(2\pi/5) & \sin(2\pi/5) \\ 0 & -\sin(2\pi/5) & \cos(2\pi/5) \end{pmatrix}, M_2 = \frac{1}{\sqrt{5}} \begin{pmatrix} 1 & 2 & 0 \\ -2 & 1 & 0 \\ 0 & 0 & -\sqrt{5} \end{pmatrix}.$$

`MAGMA` Implementation:

```
K<y> := CyclotomicField(20);
z := y^4;
i := y^5;
M1 := Matrix(K,3,3, [1,0,0,  0,(z+z^4)/2,-i*(z-z^4)/2,  0,-i*(z^4-z)/2,(z+z^4)/2]);
M2 := 1/(2*z+2*z^4+1)*Matrix(K,3,3, [1,2,0,  2,-1,0,  0,0,-(2*z+2*z^4+1)]);
G := MatrixGroup<3,K | M1,M2>;
```

`MAGMA` calculates that the smallest degree semi-invariant polynomial is $x^2+y^2+z^2$, and that the next invariant is in degree 4. The only degree 4 invariant is the square of the polynomial above.

3.2. **The Klein group $\mathrm{PSL}_2(\mathbb{F}_7)$.** In appropriate coordinates, the generators for the subgroup $\mathrm{PSL}_2(\mathbb{F}_7) \subset \mathrm{GL}_3(\mathbb{C})$ are

$$M_1 = \begin{pmatrix} \epsilon^4 & 0 & 0 \\ 0 & \epsilon^2 & 0 \\ 0 & 0 & \epsilon \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 0 & 1 \\ 1 & 0 & 0 \\ 0 & 1 & \epsilon \end{pmatrix}, M_3 = \alpha \begin{pmatrix} \epsilon-\epsilon^6 & \epsilon^2-\epsilon^5 & \epsilon^4-\epsilon^3 \\ \epsilon^2-\epsilon^5 & \epsilon^4-\epsilon^3 & \epsilon-\epsilon^6 \\ \epsilon^4-\epsilon^3 & \epsilon-\epsilon^6 & \epsilon^2-\epsilon^5 \end{pmatrix},$$

where ϵ is a primitive 7th root of unity and $\alpha = \frac{1}{\sqrt{-7}}$.

`MAGMA` Implementation:

```
K<z> := CyclotomicField(7);
M1 := Matrix(K,3,3, [z^4,0,0,  0,z^2,0, 0,0,z]);
M2 := Matrix(K,3,3, [0,0,1,  1,0,0, 0,1,0]);
M3 := -1/(2*z^4+2*z^2+2*z+1)*Matrix(K,3,3, [z-z^6,z^2-z^5,z^4-z^3,
z^2-z^5,z^4-z^3,z-z^6,   z^4-z^3,z-z^6,z^2-z^5]);
G := MatrixGroup<3,K | M1,M2,M3>;
```

`MAGMA` confirms that the smallest degree semi-invariant polynomial is the Klein quartic $x^3y+y^3z+z^3x=0$.

3.3. **The Valentiner Group $3.A_6$.** In appropriate coordinates, the generators for the subgroup $3.A_6 \subset \mathrm{GL}_3(\mathbb{C})$ are

$$M_1 = \begin{pmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 1 & 0 & 0 \end{pmatrix}, M_3 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 0 & \zeta^2 \\ 0 & -\zeta & 0 \end{pmatrix}, M_4 = \frac{1}{2} \begin{pmatrix} 1 & \tau & \tau^{-1} \\ \tau^{-1} & \tau & 1 \\ \tau & -1 & \tau^{-1} \end{pmatrix},$$

where ζ is a primitive third root of unity and $\tau = \frac{1+\sqrt{5}}{2}$.

MAGMA Implementation:

```
K<x> := CyclotomicField(15);
y := x^3;
z := x^5;
M1 := Matrix(K,3,3, [-1,0,0,  0,1,0,  0,0,-1]);
M2 := Matrix(K,3,3, [0,0,1,  1,0,0, 0,1,0]);
M3 := Matrix(K,3,3, [1,0,0,  0,0,z^2, 0,-z,0]);
M4 := 1/2*Matrix(K,3,3, [1,1/(y+y^4+1),-(y+y^4+1), 1/(y+y^4+1),(y+y^4+1),1,
(y+y^4+1),-1,1/(y+y^4+1)]);
G := MatrixGroup<3,K | M1,M2,M3,M4>;
```

MAGMA confirms that the smallest degree invariant polynomial is a unique polynomial of degree 6 (the next smallest invariant polynomial has degree 12).

3.4. **The Hessian Group and its Subgroups.**

The Hessian subgroup of $\mathrm{GL}_3(\mathbb{C})$ is the group generated by the following matrices.

$$M_1 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \omega & 0 \\ 0 & 0 & \omega^2 \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 1 & 0 & 0 \end{pmatrix}, M_3 = \frac{1}{i\sqrt{3}} \begin{pmatrix} 1 & 1 & 1 \\ 1 & \omega & \omega^2 \\ 1 & \omega^2 & \omega \end{pmatrix}, M_4 = \begin{pmatrix} \epsilon & 0 & 0 \\ 0 & \epsilon & 0 \\ 0 & 0 & \epsilon\omega \end{pmatrix},$$

where ω is a primitive third root of unity and ϵ is a primitive 9th root of unity with $\epsilon^3 = \omega^2$.

MAGMA Implementation:

```
K<z> := CyclotomicField(9);
M1 := Matrix(K,3,3,  [1,0,0, 0,z^6,0, 0,0,z^3]);
M2 := Matrix(K,3,3,  [0,1,0, 0,0,1,   1,0,0]);
M3 := Matrix(K,3,3,  [z,0,0,  0,z,0,  0,0,z^7]);
M4 := Matrix(K,3,3,  [1,1,1,  1,z^6,z^3,  1,z^3,z^6])*1/(2*z^6+1);
G := MatrixGroup<3,K| M1,M2,M3,M4>;
```

The group G has semidirect product structure $G \cong H_3 \ltimes \mathrm{SL}_2(\mathbb{F}_3)$, where the Heisenberg group H_3 is the group of order 27 generated by M_1, M_2, and ωI_3.

There are three primitive subgroups of the Hessian group G: G itself, a group H_1 of index 3, and a group H_2 of index 6 (from Blichfeldt). We can find these in MAGMA using the "Subgroups" function. MAGMA confirms that the smallest degree semi-invariant polynomial has degree 6 for each of these three groups.

3.5. **Exceptional Hypersurfaces Constructed Using $N = 3$ Groups.**

Example 3.1. Let $X \subset \mathbb{P}^2$ be the degree 4 curve defined by

$$x_0^3 x_1 + x_1^3 x_2 + x_2^3 x_0 = 0.$$

We define the following 3×3 matrices:

$$M_1 = i\begin{pmatrix}\epsilon^4 & 0 & 0\\ 0 & \epsilon^2 & 0\\ 0 & 0 & \epsilon\end{pmatrix}, M_2 = \begin{pmatrix}0 & 0 & 1\\ 1 & 0 & 0\\ 0 & 1 & \epsilon\end{pmatrix}, M_3 = \alpha\begin{pmatrix}\epsilon-\epsilon^6 & \epsilon^2-\epsilon^5 & \epsilon^4-\epsilon^3\\ \epsilon^2-\epsilon^5 & \epsilon^4-\epsilon^3 & \epsilon-\epsilon^6\\ \epsilon^4-\epsilon^3 & \epsilon-\epsilon^6 & \epsilon^2-\epsilon^5\end{pmatrix},$$

where ϵ is a primitive 7th root of unity and $\alpha = \frac{1}{\sqrt{-7}}$.

Let the group $G \subset \mathrm{GL}_3(\mathbb{C})$ be generated by M_1, M_2, M_3.

MAGMA Implementation:

```
K<y> := CyclotomicField(28);
z := y^4;
i := y^7;
M1 := i*Matrix(K,3,3, [z^4,0,0,  0,z^2,0, 0,0,z]);
M2 := Matrix(K,3,3, [0,0,1,  1,0,0, 0,1,0]);
M3 := -1/(2*z^4+2*z^2+2*z+1)*Matrix(K,3,3, [z-z^6,z^2-z^5,z^4-z^3,
z^2-z^5,z^4-z^3,z-z^6,   z^4-z^3,z-z^6,z^2-z^5]);
G := MatrixGroup<3,K | M1,M2,M3>;
```

This group G has order 648 and equals $\mathrm{Lin}(f)$, where f is the polynomial above defining the curve. (Note that the only difference between this and the $\mathrm{PSL}_2(\mathbb{F}_7)$ example is the central extension by μ_4.)

Example 3.2. Let $X \subset \mathbb{P}^2$ be the degree 6 curve defined by

$$10x_0^3x_1^3 + 9(x_0^5 + x_1^5)x_2 - 45x_0^2x_1^2x_2^2 - 135x_0x_1x_2^4 + 27x_2^6 = 0.$$

We define the following 3×3 matrices:

$$M_1 = \begin{pmatrix}1 & 0 & 0\\ 0 & -1 & 0\\ 0 & 0 & 1\end{pmatrix}, M_2 = \begin{pmatrix}0 & 1 & 0\\ 0 & 0 & 1\\ 1 & 0 & 0\end{pmatrix}, M_3 = \begin{pmatrix}1 & 0 & 0\\ 0 & 0 & \zeta^2\\ 0 & -\zeta & 0\end{pmatrix}, M_4 = \frac{1}{2}\begin{pmatrix}1 & \tau & \tau^{-1}\\ \tau^{-1} & \tau & 1\\ \tau & -1 & \tau^{-1}\end{pmatrix},$$

where ζ is a primitive third root of unity and $\tau = \frac{1+\sqrt{5}}{2}$.

Let the group $G \subset \mathrm{GL}_3(\mathbb{C})$ be generated by M_1, M_2, M_3, M_4.

MAGMA Implementation:

```
K<x> := CyclotomicField(15);
y := x^3;
z := x^5;
M1 := Matrix(K,3,3, [1,0,0,  0,-1,0,  0,0,1]);
M2 := Matrix(K,3,3, [0,0,1,  1,0,0, 0,1,0]);
M3 := Matrix(K,3,3, [1,0,0,  0,0,z^2, 0,-z,0]);
M4 := 1/2*Matrix(K,3,3, [1,1/(y+y^4+1),-(y+y^4+1), 1/(y+y^4+1),(y+y^4+1),1,
(y+y^4+1),-1,1/(y+y^4+1)]);
G := MatrixGroup<3,K | M1,M2,M3,M4>;
```

This group G has order 2160 and equals $\mathrm{Lin}(f)$, where f is equal to the polynomial above after a change change of coordinates. (Note that the only difference between this and the $3.A_6$ example is the central extension by μ_2.) We don't write the matrix generators in the coordinates for the equation above because the matrices become very messy in these coordinates.

4. $N = 4$

4.1. **Tensor Products of $N = 2$ Groups.** One way to construct subgroups of $\mathrm{GL}_4(\mathbb{C})$ is to take tensor products of subgroups of $\mathrm{GL}_2(\mathbb{C})$ as follows. For two matrices

$$A = \begin{pmatrix} a & b \\ c & d \end{pmatrix}, B = \begin{pmatrix} e & f \\ g & h \end{pmatrix},$$

the tensor product is

$$\begin{pmatrix} ae & af & be & bf \\ ag & ah & bg & bh \\ ce & cf & de & df \\ cg & ch & df & dh \end{pmatrix}.$$

In this way, we can begin with $G_1 \subset \mathrm{GL}_2(\mathbb{C})$ and $G_2 \subset \mathrm{GL}_2(\mathbb{C})$ and take tensor products of all pairs of matrix generators to get a map $G_1 \times G_2 \to \mathrm{GL}_4(\mathbb{C})$. This map is almost injective, except it takes $(-I_2, -I_2) \mapsto I_4$, so for the primitive subgroup pairs of $\mathrm{GL}_2(\mathbb{C})$, it will always have kernel of order 2. Blichfeldt classifies the subgroups that arise this way, and some extensions of these groups, on pg. 165-169 of his book. In his numbering, the groups with $[G : Z(G)] \geq 648$ are

- Group $4°$: $((2.A_4) \times (2.A_5))/\mu_2$, $[G : Z(G)] = 720$
- Group $6°$: $((2.S_4) \times (2.A_5))/\mu_2$, $[G : Z(G)] = 1440$
- Group $7°$: $((2.A_5) \times (2.A_5))/\mu_2$, $[G : Z(G)] = 3600$
- Group $11°$: Extension of case 7 with $[G : Z(G)] = 7200$
- Group $12°$: Extension of $((2.S_4) \times (2.S_4))/\mu_2$ with $[G : Z(G)] = 1152$

Here's the `MAGMA` implementation of all these groups:

```
//Tensor products of GL_2(C) groups:
//Generators for binary icosahedral subgroup of GL2(C)
K<y> := CyclotomicField(40);
z := y^8;
M1 := Matrix(K,2,2,[-z^3,0, 0,-z^2]);
M2 := 1/(2*z + 2*z^4 + 1)*Matrix(K,2,2,[-z+z^4, z^2-z^3,  z^2-z^3, z-z^4]);
I := MatrixGroup<2,K | M1,M2>;
t := y^5;
N1 := 1/2*Matrix(K,2,2,[1+t^2,1+t^2,  -1+t^2,1-t^2]);
N2 := Matrix(K,2,2,[0,1, -1,0]);
N3 := Matrix(K,2,2, [t,0, 0,t^7]);
O := MatrixGroup<2,K | N1,N2,N3>; //binary octahedral group 2.S4
T := MatrixGroup<2,K | N1,N2>; //binary tetrahedral group 2.A4
```

```
B1 := TensorProduct(M1,M1);
B2 := TensorProduct(M1,M2);
B3 := TensorProduct(M2,M1);
B4 := TensorProduct(M2,M2);
B5 := TensorProduct(N1,N1);
B6 := TensorProduct(N1,N2);
B7 := TensorProduct(N1,N3);
B8 := TensorProduct(N2,N1);
B9 := TensorProduct(N2,N2);
B10 := TensorProduct(N2,N3);
B11 := TensorProduct(N3,N1);
B12 := TensorProduct(N3,N2);
B13 := TensorProduct(N3,N3);
B14 := TensorProduct(M1,N1);
B15 := TensorProduct(M1,N2);
B16 := TensorProduct(M1,N3);
B17 := TensorProduct(M2,N1);
B18 := TensorProduct(M2,N2);
B19 := TensorProduct(M2,N3);
//Now I'll construct those groups in Blichfeldt Sections 121 and 122
//which have center of index at least 648
//4 in his notation, with [G4:Z(G4)] = 720
G4 := MatrixGroup<4,K | B14,B15,B17,B18>;
//6 in his notation, with [G6:Z(G6)] = 1440
G6 := MatrixGroup<4,K | B14,B15,B16,B17,B18,B19>;
//7 in his notation, with [G7:Z(G7)] = 3600
G7 := MatrixGroup<4,K | B1,B2,B3,B4>;
//11 in his notation, with [G11:Z(G11)] = 7200
T1 := t*Matrix(K,4,4, [1,0,0,0,  0,0,1,0,  0,1,0,0,  0,0,0,1]);
G11 := MatrixGroup<4,K | B1,B2,B3,B4,T1>;
//12 in his notation, with [G12:Z(G12)] = 1152
G12 := MatrixGroup<4,K | B5,B6,B7,B8,B9,B10,B11,B12,B13,T1>;
```

As Blichfeldt notes on pg. 169, all these groups leave the quadric $x_0x_3 - x_1x_2 = 0$ invariant, so this gives a semi-invariant polynomial for each group of degree 2. However, the only semi-invariant polynomials of these groups are powers of this quadric for $d < 12$ for $4^\circ, 6^\circ, 7^\circ, 11^\circ$, and for $d < 6$ for 12° (this can be verified with `MAGMA`).

4.2. **The group $2.A_7$.** For fun, here are the generators of $2.A_7 \subset \mathrm{GL}_4(\mathbb{C})$ (from Blichfeldt):

$$M_1 = \begin{pmatrix}1 & 0 & 0 & 0\\ 0 & \beta & 0 & 0\\ 0 & 0 & \beta^4 & 0\\ 0 & 0 & 0 & \beta^2\end{pmatrix}, M_2 = \begin{pmatrix}1 & 0 & 0 & 0\\ 0 & 0 & 1 & 0\\ 0 & 0 & 0 & 1\\ 0 & 1 & 0 & 0\end{pmatrix}, M_3 = \frac{1}{\sqrt{-7}}\begin{pmatrix}p^2 & 1 & 1 & 1\\ 1 & -q & -p & -p\\ 1 & -p & -q & -p\\ 1 & -p & -p & -q\end{pmatrix},$$

where β is a primitive 7th root of unity, $p = \beta + \beta^2 + \beta^4$, and $q = \beta^3 + \beta^5 + \beta^6$.

MAGMA Implementation:

```
K<z> := CyclotomicField(7);
sqrtneg7 := 1 + 2*z + 2*z^2 + 2*z^4;
p := z + z^4 + z^2;
q := z^3 + z^5 + z^6;
M1 := Matrix(K,4,4,[1,0,0,0, 0,z,0,0, 0,0,z^4,0, 0,0,0,z^2]);
M2 := Matrix(K,4,4,[1,0,0,0, 0,0,1,0, 0,0,0,1, 0,1,0,0]);
M3 := 1/sqrtneg7*Matrix(K,4,4, [p^2,1,1,1, 1,-q,-p,-p, 1,-p,-q,-p, 1,-p,-p,-q]);
G := MatrixGroup<4,K|M1,M2,M3>;
```

MAGMA and GAP both confirm that the minimum degree of a polynomial semi-invariant (same as invariant here) is 8.

4.3. The group $\mathrm{Sp}_4(\mathbb{F}_3)$.

Here are the generators of $\mathrm{Sp}_4(\mathbb{F}_3) \subset \mathrm{GL}_4(\mathbb{C})$ (from Blichfeldt):

$$M_1 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \end{pmatrix}, M_2 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \omega & 0 \\ 0 & 0 & 0 & \omega^2 \end{pmatrix}, M_3 = \begin{pmatrix} \omega & 0 & 0 & 0 \\ 0 & \omega & 0 & 0 \\ 0 & 0 & \omega & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix},$$
$$M_4 = \frac{1}{\sqrt{-3}} \begin{pmatrix} \sqrt{-3} & 0 & 0 & 0 \\ 0 & 1 & 1 & 1 \\ 0 & 1 & \omega & \omega^2 \\ 0 & 1 & \omega^2 & \omega \end{pmatrix}, M_5 = \begin{pmatrix} 0 & 0 & -1 & 0 \\ 0 & 1 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix},$$

where ω is a primitive 3rd root of unity.

MAGMA Implementation:

```
K<z> := CyclotomicField(3);
k := (z^2 - z)/3;
M1 := Matrix(K,4,4,[1,0,0,0, 0,0,1,0, 0,0,0,1, 0,1,0,0]);
M2 := Matrix(K,4,4,[1,0,0,0, 0,1,0,0, 0,0,z,0, 0,0,0,z^2]);
M3 := Matrix(K,4,4,[z,0,0,0, 0,z,0,0, 0,0,z,0, 0,0,0,1]);
M4 := k*Matrix(K,4,4,[1/k,0,0,0, 0,1,1,1, 0,1,z,z^2, 0,1,z^2,z]);
M5 := Matrix(K,4,4,[0,0,-1,0, 0,1,0,0, -1,0,0,0, 0,0,0,-1]);
G := MatrixGroup<4,K|M1,M2,M3,M4,M5>;
```

MAGMA confirms our previous GAP calculation that the minimum degree of a polynomial semi-invariant (same as invariant here) is 12.

4.4. Normalizer of the Extra-special Group.

The remaining case is the normalizer N of the extra-special group H_4 of order 32 in $\mathrm{GL}_4(\mathbb{C})$. This group fits into an exact sequence

$$0 \to \tilde{H}_4 \to N \to S_6 \to 0.$$

Here $\tilde{H}_4$ is the product of H_4 with μ_2. The matrix generators for H_4 are

$$M_1 = \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix}, M_2 = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}, M_3 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, M_4 = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}.$$

To obtain the entire group N we add the two additional generators

$$M_5 = \frac{1+i}{\sqrt{2}} \begin{pmatrix} i & 0 & 0 & 0 \\ 0 & i & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, M_6 = \frac{1+i}{2} \begin{pmatrix} -i & 0 & 0 & i \\ 0 & 1 & 1 & 0 \\ 1 & 0 & 0 & 1 \\ 0 & -i & i & 0 \end{pmatrix}.$$

MAGMA Implementation:

```
K<z> := CyclotomicField(8);
M1 := Matrix(K,4,4,[0,0,1,0,  0,0,0,1,  1,0,0,0,  0,1,0,0]);
M2 := Matrix(K,4,4,[0,1,0,0,  1,0,0,0,  0,0,0,1,  0,0,1,0]);
M3 := Matrix(K,4,4,[1,0,0,0,  0,-1,0,0, 0,0,-1,0,  0,0,0,1]);
M4 := Matrix(K,4,4,[1,0,0,0,  0,1,0,0,  0,0,-1,0, 0,0,0,-1]);
M5 := z*Matrix(K,4,4, [z^2,0,0,0, 0,z^2,0,0, 0,0,1,0, 0,0,0,1]);
M6 := (1+z^2)/2*Matrix(K,4,4, [-z^2,0,0,z^2, 0,1,1,0, 1,0,0,1, 0,-z^2,z^2,0]);
G := MatrixGroup<4,K|[M1,M2,M3,M4,M5,M6]>;
```

There are several subgroups of this group which are primitive: Blichfeldt lists them as 13° through 21° on pg. 172 of his book. We'll only be concerned with 16° through 21° since these have $[G : Z(G)] \geq 648$.

We can search through the different subgroups of the group above using MAGMA's "subgroups" function. The calculation gives that all of these have semi-invariants of degree at least 4, but mostly starting only in degree 8, with two exceptions! The largest exception (and the only one which gives an exceptional example) is 19°, which has a semi-invariant of degree 4. This group is generated by M_1, M_2, M_3, M_4, M_6 above and

$$M_8 = \frac{1+i}{\sqrt{2}} \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}.$$

(The numbering is because I also coded group 18° to check it.) To compute the group 19° in MAGMA use (in addition to previous code):

```
M7 := z*Matrix(K,4,4,[1,0,0,0, 0,z^2,0,0, 0,0,z^2,0, 0,0,0,1]);
M8 := z*Matrix(K,4,4,[1,0,0,0, 0,1,0,0, 0,0,1,0, 0,0,0,-1]);
G18 := MatrixGroup<4,K|[M1,M2,M3,M4,M6,M7]>; // Group 18^o on Blichfeldt pg. 172
G19 := MatrixGroup<4,K|[M1,M2,M3,M4,M6,M8]>; // Group 19^o on Blichfeldt pg. 172
//This last one is Lin(f) for an exceptional example of degree 4
```

MAGMA gives that there is a semi-invariant polynomial of degree 4:

$$x_0^4 + x_1^4 + x_2^4 + x_3^4 - 6(x_0^2x_1^2 + x_0^2x_2^2 + x_0^2x_3^2 + x_1^2x_2^2 + x_1^2x_3^2 + x_2^2x_3^2).$$

Since the center is already μ_4, this is already $\mathrm{Lin}(f)$ for f the equation above, which is another exception because $|\mathrm{Lin}(f)| = 7860 > 24 \cdot 4^4$.

5. $N = 6$

For $N = 6$, there are only two groups large enough that we need to analyze. One of these, a central extension of the Janko group J_2, we can analyze using GAP (use `AtlasGroup("2.J2")`) since J_2 is a simple group. This gives that it has smallest degree semi-invariant 12.

The other group is the largest possible in dimension 6, which Collins writes as $6_1.U_4(3).2_2$. A paper by Lindsay (cited in main text) gives the matrix generators for this group. Indeed, this group is generated by all 6×6 permutation matrices, all order 3 matrices of determinant 1, and the matrix $I_6 - Q/3$, where Q is the 6×6 matrix with all entries equal to 1.

MAGMA Implementation:

```
//Construction of largest primitive subgroup of GL_6(C); source is
//Lindsay's "On a six-dimensional projective representation of PSU_4(3)"
K<z> := CyclotomicField(3);
M1 := Matrix(K,6,6,[0,1,0,0,0,0, 1,0,0,0,0,0, 0,0,1,0,0,0,
0,0,0,1,0,0, 0,0,0,0,1,0, 0,0,0,0,0,1]);
M2 := Matrix(K,6,6,[0,1,0,0,0,0, 0,0,1,0,0,0, 0,0,0,1,0,0,
0,0,0,0,1,0, 0,0,0,0,0,1, 1,0,0,0,0,0]);
M3 := DiagonalMatrix(K,6, [z,1,1,1,1,z^2]);
M4 := DiagonalMatrix(K,6, [1,z,1,1,1,z^2]);
M5 := DiagonalMatrix(K,6, [1,1,z,1,1,z^2]);
M6 := DiagonalMatrix(K,6, [1,1,1,z,1,z^2]);
M7 := DiagonalMatrix(K,6, [1,1,1,1,z,z^2]);
M8 := 1/3*Matrix(K,6,6, [2,-1,-1,-1,-1,-1, -1,2,-1,-1,-1,-1,
-1,-1,2,-1,-1,-1, -1,-1,-1,2,-1,-1, -1,-1,-1,-1,2,-1, -1,-1,-1,-1,-1,2]);
G := MatrixGroup<6,K|[M1,M2,M3,M4,M5,M6,M7,M8]>;
```

MAGMA confirms that the smallest degree semi-invariant polynomial for this group has degree 6. The unique semi-invariant polynomial of this degree is:

$$\sum_{i=0}^{5} x_i^6 - 10 \sum_{0 \le i < j \le 5} x_i^3 x_j^3 - 180 x_0 x_1 x_2 x_3 x_4 x_5.$$

The group constructed above already has center of order 6, so it equals $\mathrm{Lin}(f)$ for the polynomial f above. This gives a new exceptional example!

6. $N = 8$

For $N = 8$, there are again only two groups large enough to require study, and it turns out neither of them will yield exceptional examples. All we really need for our paper is that the groups have semi-invariant polynomials only in even degree. For explicitness, here is a description of the largest group $2.O_8^+(2).2$: it is the group generated by all 8×8 permutation matrices, all 8×8 matrices of determinant 1 and order 2, and the matrix $I_8 - P/4$, where P is the matrix with all entries equal to 1.

MAGMA Implementation:

```
//Construction of largest primitive subgroup of GL_8(C); source is
//Lindsay's "On a six-dimensional projective representation of PSU_4(3)"
K := Rationals();
M1 := Matrix(K,8,8,[0,1,0,0,0,0,0,0, 1,0,0,0,0,0,0,0, 0,0,1,0,0,0,0,0,
 0,0,0,1,0,0,0,0, 0,0,0,0,1,0,0,0, 0,0,0,0,0,1,0,0, 0,0,0,0,0,0,1,0, 0,0,0,0,0,0,0,1]);
M2 := Matrix(K,8,8,[0,1,0,0,0,0,0,0, 0,0,1,0,0,0,0,0, 0,0,0,1,0,0,0,0,
 0,0,0,0,1,0,0,0, 0,0,0,0,0,1,0,0, 0,0,0,0,0,0,1,0, 0,0,0,0,0,0,0,1, 1,0,0,0,0,0,0,0]);
M3 := DiagonalMatrix(K,8, [-1,1,1,1,1,1,1,-1]);
M4 := DiagonalMatrix(K,8, [1,-1,1,1,1,1,1,-1]);
M5 := DiagonalMatrix(K,8, [1,1,-1,1,1,1,1,-1]);
M6 := DiagonalMatrix(K,8, [1,1,1,-1,1,1,1,-1]);
M7 := DiagonalMatrix(K,8, [1,1,1,1,-1,1,1,-1]);
M8 := DiagonalMatrix(K,8, [1,1,1,1,1,-1,1,-1]);
M9 := DiagonalMatrix(K,8, [1,1,1,1,1,1,-1,-1]);
M10 := 1/4*Matrix(K,8,8, [3,-1,-1,-1,-1,-1,-1,-1, -1,3,-1,-1,-1,-1,-1,-1,
-1,-1,3,-1,-1,-1,-1,-1, -1,-1,-1,3,-1,-1,-1,-1, -1,-1,-1,-1,3,-1,-1,-1,
-1,-1,-1,-1,-1,3,-1,-1, -1,-1,-1,-1,-1,-1,3,-1, -1,-1,-1,-1,-1,-1,-1,3]);
G := MatrixGroup<8,K|[M1,M2,M3,M4,M5,M6,M7,M8,M9,M10]>;
```

MAGMA confirms that this only has semi-invariants of even degree.

DEPARTMENT OF MATHEMATICS, PRINCETON UNIVERSITY, FINE HALL, WASHINGTON ROAD, PRINCETON, NJ 08544-1000, USA

Email address: esserl@math.princeton.edu

Email address: jenniferli@princeton.edu